# Simulation

*This lecture describes how to use Verilog to simulate designs.*
*After this lecture you should be able to write a testbench that can: set initial values, generate clocks, read test vectors from a file, display values, and terminate on a condition.*

## Simulation

HDLs can be used to test hardware designs by simulating their operation. These simulations consists of the module being tested (called the Design Under Test or *DUT*) that is instantiated within another module called a *testbench*. The testbench applies inputs to the DUT and checks its outputs:




## Test Vectors

The inputs to the DUT and the corresponding expected outputs are called test vectors. These can be generated by the testbench itself or they can be read from a file.

Test vectors should include:

1. typical inputs,
2. minimum and maximum valid inputs,
3. invalid inputs, and
4. randomly-chosen values.

**Exercise 1:** Give examples of appropriate test inputs for each of the above categories if you were testing a circuit that computed the square root of a 16-bit signed number.

## Verilog for Verification

The following Verilog language features are useful for simulation but many can't be implemented in hardware ("synthesized").

## Blocks

A block of code is a sequence of one or more statements. In Verilog if a block contains more than one statement then it is delimited with **`begin`** and **`end`** keywords. The C language also has blocks, but the delimiters are braces (`{...}`).

## `initial` and `always` blocks

**`initial`** and **`always`** statements control the execution of a block similar to the way an **`if`** statement controls execution of a block. A module may have any number of **`initial`** and **`always`** blocks.

**`initial`** blocks execute once at the start of the simulation and are used often used to initialise signals.

**`always`** blocks, for our purposes, execute continuously again and again[1]

## `if/else/for/while`

**`if`**/**`else`**, **`for`**, and **`while`** statements, whose syntax and semantics are similar to those C, can be used within **`initial`** or **`always`** blocks. However, in HDL simulations control flow is often based on conditions or events as described below.

## Delays

Placing `#`*number* before a statement delays execution by *number* simulation time. The suffixes **`ns`** and **`us`** can be used for nano- and micro-seconds. Delays are *not* synthesizable because they cannot be easily implemented in hardware.

Placing `@(`*event*`)` before a statement where *event* where can be a signal name, optionally with a **`posedge`** or **`negedge`** prefix, delays execution until that signal changes or the appropriate edge.

[1] Specifically, execution of the block starts at every "time step".

### wait

The **wait(***expression***)** statement pauses until the expression is true (non-zero).

**Exercise 2:** What's the difference between: **always @(x) y = '1;**, **wait(x) y='1;**, and **@(x) y='1;**?

## System Tasks

Functions beginning with **$** are called system tasks. Useful ones include:

- **$write()** is similar to C's **printf()**, and can be used to print values during a simulation;
- **$dumpfile** and **$dumpvars** record changes in signals to a **.vcd** file for subsequent viewing with a waveform viewer.
- **$fopen()** and **$fscanf()**, similar to the C library functions **fopen()** and **fscanf()**, can open and read from text files.
- **$finish** and **$stop** terminate or suspend a simulation.

There are many other system functions. For example, to obtain the dimensions or limits of arrays, compute math functions, return the current simulation time, and initialize arrays from files. None are synthesizable.

## Example

The testbench below demonstrates the language features described above.

The example DUT is a module with an 8-bit input that outputs a 16-bit sum of all the odd-valued inputs since a reset input was asserted.

An **initial** block opens a file containing test vectors, initializes the clock signal, and **$stop**'s the simulation at the end of the test vector file.

An **always** block inverts the clock every 0.5 $\mu$s; this creates a 1 MHz clock.

The final **always** block continuously reads and applies the input(s) from each test vector, waits for the falling edge of the clock, and compares the DUT output(s) to the value(s) read from the test vectors.

The example testbench checks the DUT outputs just before the falling edge of the clock and changes the DUT inputs just after the falling edge:




This offsets changes in the inputs and outputs by half a clock period from the rising clock edge to make the waveforms easier to understand. This is a *functional* simulation which means the delay of the DUT is not being checked[2]

Note that the simulation code accesses the value of **out** within the DUT using the notation <*instance name*>**.**<*signal name*> (**ex66_0.out** in this case). This violates the principles of modularity and "data hiding" but can be useful for troubleshooting.

The test vectors are read from the file **ex66data.csv** containing the following lines[3]:

```
0,0
1,1
2,1
3,4
4,4
5,8
7,16
```

Running this testbench using the Modelsim simulation program creates the waveform files shown in Figure 1. The following lines are printed showing the inputs, actual output, and the expected output:

```
# run -all
#       in  out_tb     out
#        0       0       0
#        1       1       1
#        2       1       1
#        3       4       4
#        4       4       4
#        5       8       9 *** Error
#        7      16      16
# ** Note: $stop    : ex66.sv(61)
#    Time: 7 us  Iteration: 1  Instance: /ex66_tb
# Break in Module ex66_tb at ex66.sv line 61
```

**Exercise 3:** How could you: (a) terminate the simulation if a test vector failed? (b) change the clock frequency to 10 MHz? (c) print each test vector as it's read? (d) assert the reset input for two clock cycles?

**Exercise 4:** What statements could you use in an initial block to create this waveform on the signal **x**?

[2] If the DUT description includes delays then a simulation can also check that the DUT operates correctly at the applied clock rate.

[3] This file is in Comma Separated Values (.csv) format.




Figure 1: Simulation waveforms (from **`ex66.vcd`**).




```
// output cumulative sum of odd-valued inputs

module ex66
  ( input logic reset, clk,
    input logic [7:0] in,
    output logic [15:0] out ) ;

   always_ff @(posedge clk)
     out <= reset ? '0 :
            in & 1 ? out+in : out ;

endmodule

// example testbench

module ex66_tb ;

   // DUT inputs and outputs
   logic reset, clk ;
   logic [7:0] in ;
   logic [15:0] out, out_tb ;

   // instantiate DUT
   ex66 ex66_0 (.*) ;

   // file descriptor and number of values by fscanf()
   integer fd, n ;

   // initialization
   initial begin

      // record all signals in a .vcd file
      $dumpfile("ex66.vcd") ;
      $dumpvars ;

      // initialize DUT signals and simulation variables
      reset = '1 ;
      clk = '0 ;

      // open test vector file
      fd = $fopen("ex66data.csv","r") ;

      // print header
      $write("%8s%8s%8s\n", "in", "out_tb", "out") ;

      // de-assert reset after first rising edge
      @(negedge clk) reset = '0 ;

   end

   // generate 1 MHz clock (2x500ns period)
   always #0.5us clk = ~clk ;

   // continuously apply input, wait, & check output
   always begin

      // read (and apply) DUT inputs
      n = $fscanf(fd,"%d,%d", in, out_tb) ;

      // stop if didn't read 2 values
      if ( n != 2 ) $stop ;

      //  display test vector
      $write("%8d%8d",in,out_tb) ;

      // wait for falling clock edge
      @(negedge clk) ;

      // show DUT state and check output
      $write( "%8d%s\n", ex66_0.out,
              out != out_tb ? " *** Error" : "" ) ;

   end

endmodule
```

**Exercise 5:** Write a testbench for the traffic light controller in a previous lecture that asserts reset for one clock cycle, waits for the lights to change 5 times, waits 3 clock cycles, and then terminates.